# Maintainer: AndyRTR <andyrtr@archlinux.org>
# Contributor: Simone Sclavi 'Ito' <darkhado@gmail.com>

pkgname=libmspub
pkgver=0.1.4
pkgrel=3
pkgdesc="Microsoft Publisher file format parser library (latest snapshot)"
arch=('x86_64')
url="https://wiki.documentfoundation.org/DLP/Libraries/libmspub"
license=('GPL2' 'LGPL2.1' 'MPL')
depends=('libwpd' 'icu' 'librevenge')
makedepends=('libwpg' 'boost' 'doxygen')
source=(https://dev-www.libreoffice.org/src/${pkgname}/${pkgname}-${pkgver}.tar.xz)
sha256sums=('ef36c1a1aabb2ba3b0bedaaafe717bf4480be2ba8de6f3894be5fd3702b013ba')

build() {
    cd ${pkgname}-${pkgver}
    ./configure --prefix=/usr --with-docs
    make
}

package() {
    cd ${pkgname}-${pkgver}
    make DESTDIR=$pkgdir install
}
